;;; Mudsync --- Live hackable MUD
;;; Copyright © 2016 Christopher Allan Webber <cwebber@dustycloud.org>
;;;
;;; This file is part of Mudsync.
;;;
;;; Mudsync is free software; you can redistribute it and/or modify it
;;; under the terms of the GNU General Public License as published by
;;; the Free Software Foundation; either version 3 of the License, or
;;; (at your option) any later version.
;;;
;;; Mudsync is distributed in the hope that it will be useful, but
;;; WITHOUT ANY WARRANTY; without even the implied warranty of
;;; MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the GNU
;;; General Public License for more details.
;;;
;;; You should have received a copy of the GNU General Public License
;;; along with Mudsync.  If not, see <http://www.gnu.org/licenses/>.

(define-module (mudsync room)
  #:use-module (mudsync command)
  #:use-module (mudsync gameobj)
  #:use-module (8sync actors)
  #:use-module (8sync agenda)
  #:use-module (oop goops)
  #:use-module (srfi srfi-1)
  #:use-module (ice-9 control)
  #:export (<room> <exit>))


;;; Exits
;;; =====

(define-class <exit> ()
  (to #:init-keyword #:to)
  ;; Name of the room (@@: Should this be names?)
  (name #:getter exit-name
        #:init-keyword #:name)
  (desc #:init-keyword #:desc
        #:init-value #f)

  ;; *Note*: These two methods have an extra layer of indirection, but
  ;;   it's for a good reason.
  (visible-check #:init-value (const #t)
                 #:init-keyword #:visible-check)
  ;; By default all exits can be traversed
  (traverse-check #:init-value (const #t)
                  #:init-keyword #:traverse-check))

(define* (exit-can-traverse? exit actor
                             #:optional (target-actor (actor-id actor)))
  ((slot-ref exit 'traverse-check) exit actor target-actor))

(define* (exit-is-visible? exit actor
                           #:optional (target-actor (actor-id actor)))
  ((slot-ref exit 'traverse-check) exit actor target-actor))



;;; Rooms
;;; =====

(define %room-contain-commands
  (list
   (loose-direct-command "look" 'cmd-look-at)
   (empty-command "look" 'cmd-look-room)
   (empty-command "go" 'cmd-go-where)
   (loose-direct-command "go" 'cmd-go)
   (greedy-command "say" 'cmd-say)
   (greedy-command "\"" 'cmd-say)
   (greedy-command "'" 'cmd-say)
   (greedy-command "emote" 'cmd-emote)
   (greedy-command "/me" 'cmd-emote)))

;; TODO: Subclass from container?
(define-class <room> (<gameobj>)
  ;; A list of <exit>
  (exits #:init-value '()
         #:init-keyword #:exits
         #:getter room-exits)

  (container-commands
   #:init-value (wrap %room-contain-commands))

  (actions #:allocation #:each-subclass
           #:init-value
           (build-actions
            (cmd-go room-cmd-go)
            (cmd-go-where room-cmd-go-where)
            (announce-entrance room-announce-entrance)
            (look-room room-look-room)
            (tell-room room-act-tell-room)
            ;; in this case the command is the same version as the normal
            ;; look-room version
            (cmd-look-room room-look-room)
            (cmd-look-at room-look-at)
            (cmd-say room-cmd-say)
            (cmd-emote room-cmd-emote))))

(define* (room-cmd-go room message #:key direct-obj)
  (define exit
    (find
     (lambda (exit)
       (equal? (exit-name exit) direct-obj))
     (room-exits room)))
  (define to-address (if exit
                         ;; Get the exit, but resolve it dynamically
                         ;; in case it's a special
                         (dyn-ref room (slot-ref exit 'to))
                         #f))
  (define player-name
    (mbody-val (<-wait (message-from message) 'get-name)))
  (cond
   (exit
    ;; Set the player's new location
    (<-wait (message-from message) 'set-loc!
            #:loc to-address)
    ;; Tell everyone else the person walked away
    (room-tell-room
     room
     (format #f "~a wanders ~a.\n"
             player-name direct-obj))
    (<- to-address 'announce-entrance
        #:who-entered (message-from message))
    ;; Have the new room update the player to the new location
    (<- to-address 'look-room
        #:to-id (message-from message)))
   (else
    (<- (message-from message) 'tell
        #:text "You don't see any way to go there.\n"))))

(define (room-cmd-go-where room message)
  (<- (message-from message) 'tell
      #:text "Go where?\n"))

;;; look commands

(define (list-words-as-string words)
  "A little utility for listing a bunch of words in an English-style list"
  ;; TODO: This could be made faster by traversing the O(n)
  ;;   list once, not twice
  (let ((word-length (length words)))
    (cond 
     ((eqv? word-length 0) "")
     ((eqv? word-length 1) (car words))
     (else
      ;; TODO: and this is NOT efficient
      (string-append
       (string-join
        (drop-right words 1)
        ", ")
       " and "
       (last words))))))

(define (room-player-looks-around room player-id)
  "Handle looking around the room"
  ;; Get the room text
  (define room-text
    `((strong "=> " ,(slot-ref room 'name) " <=")
      (p ,(slot-ref room 'desc))))

  ;; Get a list of other things the player would see in the room
  (define occupant-names-all
    (map
     (lambda (occupant)
       (call-with-message (<-wait occupant 'visible-name
                                  #:whos-looking player-id)
                          (lambda* (_ #:key text)
                            text)))
     (remove
      (lambda (x) (equal? x player-id))
      (hash-map->list (lambda (x _) x)
                      (slot-ref room 'occupants)))))

  ;; Strip out the #f responses (these aren't listed because they lack a name
  ;; or they aren't "obviously visible" to the player)
  (define occupant-names-filtered
    (filter identity occupant-names-all))

  (define occupant-names-string
    (if (eq? occupant-names-filtered '())
        #f
        (format #f "You see here: ~a.\n"
                (list-words-as-string occupant-names-filtered))))

  (define final-text
    (if occupant-names-string
        `(,@room-text
          (p (i ,occupant-names-string)))
        room-text))
  
  (<- player-id 'tell
      #:text final-text))


(define* (room-look-room room message
                            ;; Either send it to the #:to-id of the message,
                            ;; or to the sender of the message
                            #:key (to-id (message-from message)))
  "Command: Player asks to look around the room"
  (room-player-looks-around room to-id))

(define (room-find-thing-called room called-this)
  "Find something called CALLED-THIS in the room, if any."
  (call/ec
   (lambda (return)
     (for-each
      (lambda (occupant)
        (mbody-receive (_ #:key goes-by)
            (<-wait occupant 'goes-by)
          (if (member called-this goes-by)
              (return occupant))))
      (hash-map->list (lambda (key val) key)
                      (slot-ref room 'occupants)))
     #f)))

(define %formless-desc
  "You don't see anything special.")

(define* (room-look-at room message #:key direct-obj)
  "Look at a specific object in the room."
  (define matching-object
    (room-find-thing-called room direct-obj))

  (cond
   (matching-object
    (let ((obj-desc
           (mbody-val (<-wait matching-object 'get-desc
                            #:whos-looking (message-from message)))))
      (if obj-desc
          (<- (message-from message) 'tell
              #:text (string-append obj-desc "\n"))
          (<- (message-from message) 'tell
              #:text (string-append %formless-desc "\n")))))
   (else
    (<- (message-from message) 'tell
        #:text "You don't see that here, so you can't look at it.\n"))))


(define* (room-tell-room room text #:key exclude wait)
  (define who-to-tell (gameobj-occupants room #:exclude exclude))
  (for-each
   (lambda (tell-me)
     ;; @@: Does anything really care?
     (define deliver-method
       (if wait
           <-wait
           <-))
     (deliver-method tell-me 'tell
                     #:text text))
   who-to-tell))

(define* (room-act-tell-room room message #:key text exclude wait)
  "Tell the room some messages."
  (room-tell-room room text
                  #:exclude exclude
                  #:wait wait))

(define* (room-cmd-say room message #:key phrase)
  "Command: Say something to room participants."
  (define player-name
    (mbody-val (<-wait (message-from message) 'get-name)))
  (define message-to-send
    `((b "<" ,player-name ">") " " ,phrase))
  (room-tell-room room message-to-send))

(define* (room-cmd-emote room message #:key phrase)
  "Command: Say something to room participants."
  (define player-name
    (mbody-val (<-wait (message-from message) 'get-name)))
  (define message-to-send
    `((b "* " ,player-name) " " ,phrase))
  (room-tell-room room message-to-send))

(define* (room-announce-entrance room message #:key who-entered)
  (define player-name
    (mbody-val (<-wait who-entered 'get-name)))
  (define message-to-send
    (format #f "~a enters the room.\n" player-name))
  (room-tell-room room message-to-send
                  #:exclude who-entered))
